
1. [bookmark: _Toc513480008]Identification des tables et des variables

Tables d’actes affinés de codage disponibles
-ER_PHA_F (médicaments)
-ER_BIO_F (actes de biologie)
-ER_CAM_F (actes médicaux)
-ER_UCD_F (médicaments rétrocédés + en sus du GHS)
-ER_LOT_F (lots de medicaments)
-ER_TIP_F (dispositifs médicaux)


Table des prestations
ER_PRS_F
Pseudo référentiel
IR_BEN_R



2 clés de jointure



9  clés de jointure





· La table des prestations (ER_PRS_F) est centrale dans le DCIR. Les variables généralement sélectionnées dans cette table sont :
· Nature de la prestation (PRS_NAT_REF)
· Dates (par exemple dates de début (EXE_SOI_DTD) et de fin de prestation (EXE_SOI_DTF))
· variables concernant le PS exécutant ou prescripteur (par exemple spécialité médicale : PSE_SPE_COD (PS exécutant) et PSP_SPE_COD (PS prescripteur))
· CMU (Couverture Maladie Universelle Complémentaire) au moment de la prestation (BEN_CMU_TOP)
· Régime d’affiliation à l’Assurance Maladie au moment de la prestation (généré à partir d’ORG_AFF_BEN)
· Département ou commune de résidence au moment de la prestation (BEN_RES_DPT et BEN_RES_COM)

NB : la sélection des prestations peut se faire sur une valeur particulière de PRS_NAT_REF (nature de la prestation) comme par exemple ‘visite chez un médecin généraliste’ et dans ce cas la jointure s’effectue sur deux tables seulement (ER_PRS_F et IR_BEN_R).

· De la table du pseudo-référentiel des bénéficiaires (IR_BEN_R), on retient généralement les variables socio-démographiques (la table contient les informations les plus récentes) et l’identifiant unique BEN_NIR_ANO en plus des 2 clés de jointure (BEN_NIR_PSA et BEN_RNG_GEM). 
Cependant, on peut garder uniquement les variables de jointure (BEN_NIR_PSA et BEN_RNG_GEM) et conserver les variables socio-démographiques associées à des prestations déjà sélectionnées (cf programme type joint) ou récupérer les variables socio-démographiques de l’année 20aa  dans la table EXTRACTION_PATIENTS20aaTR du répertoire CONSOPAT. 
· La table d’actes affinés de codage précise la nature des remboursements liés à la prestation, comme par exemple la liste des médicaments liée à la prestation(ER_PHA_F) ou les actes médicaux (ER_CAM_F). Le document « DCIR ce qu’il faut savoir » décrit ces tables en détail.
De la table des prestations affinées, on ne retient généralement que le code affiné et la quantité correspondante

	Nom de la table 
	Code affiné
	Quantité
	Remarque
	Référentiel associé

	ER_PHA_F
	PHA_PRS_C13
	PHA_ACT_QSN
	  
	IR_PHA_R

	ER_BIO_F
	BIO_PRS_IDE
	BIO_ACT_QSN
	
	IR_BIO_R

	ER_CAM_F
	CAM_PRS_IDE
	
	Il n’y a pas de variable quantité dans la table ER_CAM_F. Prendre PRS_ACT_QTE de la table ER_PRS_F.
Pensez à prendre en compte CAM_ACT_COD (code activité) + CAM_TRT_PHA (code phase de traitement).

	-


	ER_UCD_F
	UCD_UCD_COD
	UCD_DLV_NBR
	
	-

	ER_TIP_F
	TIP_PRS_IDE
	TIP_ACT_QSN
	
	NT_LPP



La sélection des codes affinés peut se faire directement dans la requête en listant le ou les codes souhaités ou en fusionnant la table de prestations affinées avec un référentiel pour effectuer une sélection sur un groupe plus important de codes. 
Par exemple : fusionner la table de prestations affinées de pharmacie : ER_PHA_R  avec le référentiel pharmacie IR_PHA_R afin de faire une sélection sur un code ATC particulier regroupant plusieurs codes. ER_PHA_F
ER_PRS_F
IR_BEN_R
IR_PHA_R







2. [bookmark: _Toc513480009]Recommandations pour les requêtes

· La fusion des tables se fait par une procédure SQL. 

· Il est recommandé de ne garder que les variables nécessaires.
· Ne pas sélectionner systématiquement les 9 variables constituant les clés de jointure entre la table ER_PRS_F et les tables affinées,
Ne pas systématiquement sélectionner en double les variables en fonction de leur provenance. (par exemple : variable BEN_SEX_COD dans IR_BEN_R et ER_PRS_F)   

· Lorsque l’on veut sélectionner des visites ou des consultations dans ER_PRS_F, il faut supprimer les enregistrements correspondant à des majorations et des compléments en utilisant le filtre CPL_MAJ_TOP < 2. 

· Pour le moment, l’information sur les soins externes réalisés en établissements publics n’est pas exhaustive, et sauf cas particulier, il convient d’appliquer un filtre pour les exclure (DPN_QLF = 71 AND PRS_DPN_QLP = 71) 

· Les tables du DCIR sont des tables volumineuses. Les prestations alimentent le DCIR par mois de flux et non en fonction de la date d’exécution de la prestation de soins. Pour ces deux raisons, la requête peut être exécutée par mois de flux et, si l’on veut les prestations sur une année complète, la requête doit être répétée 18 fois (année entière plus les 6 mois suivants). En procédant ainsi, on connait par exemple plus de 98% des délivrances de médicaments  de l’année portées au remboursement.

· Prise en compte des régularisations.
Dans les tables de prestations ou d’actes affinés de codage il existe des lignes de régularisation. Elles correspondent à des annulations de prestations avec ou non enregistrement d’une nouvelle ligne en remplacement de la ligne annulée. La ligne de régularisation est en tout point identique à la ligne initiale mais avec une quantité d’actes négative. En pratique, il faut donc sommer les quantités des lignes identiques de prestations et ne garder que les lignes dont la somme est strictement positive.
1

3.1. [bookmark: _Toc513480011]Sélection des prestations : exemple des remboursements de médicaments

/*********************************************************************************/
/************           SELECTION DES REMBOURSEMENTS DE MEDICAMENTS                          *******/
/************                    DANS DCIR ARCHIVE ET DCIR                                                                 *******/
/*********************************************************************************/

/*********************************************************************************/
/* DONNEES en entrée 
				oravue.er_prs_f_&an ou oravue.er_prs_f
				oravue.er_pha_f_&an ou oravue.er_pha_f
				oravue.IR_PHA_R
				&table_population en entrée (facultatif)
	

/* DONNEES en sortie
				&table_sortie à définir

/* PLAN du PROGRAMME ;
	  			- Définition par l'utilisateur des paramètres suivants:
					* Choix de la librairie de données en sortie
					* options obs=0 ou MAX (respectivement pour tester le
   programme ou lancer la sélection)
					* choix des codes de médicaments
					* liste des variables à sélectionner
					* table de population à croiser (facultatif)
	  			- Macro outils
	  			- Sélection des médicaments 
				- Copie de la table de population (facultatif) dans ORAUSER
	  			- Macro de sélection des médicaments pour un mois de flux :
 CREA_TABLE
	  			- Macro de compilation des mois de flux : COMPILATION
	  			- Macro de suppression des régularisations : REGULARISATIONS
	  			- Macro TAB_MEDICAMENTS qui exécute les macros CREA_TABLE et
 COMPILATION  pour chaque mois de flux et REGULARISATION sur la
 table finale
				- L'utilisateur lance la Macro TAB_MEDICAMENTS en choisissant
				  * le mois et année de début d'exécution des soins (format aaaamm)
				  * le mois et année de fin d'exécution des soins (format aaaamm)
				  * le nom de la table de sortie
	 

/*********************************************************************************/


/*********************************************************************************/
/*****  Paramètres à changer : répertoire,  liste des variables, des codes  *****/
/*********************************************************************************/

/*déclaration de la librairie */
/* exemple : sous répertoire PGM_tmt  de l’espace partage ESP_PART*/

%LET fichier = %SYSGET (HOME)/sasdata ;  
LIBNAME ma_lib "&fichier/sasdata/ESP_PART/PGM_tmt" ;
  
OPTIONS OBS  =0 ; /*pour tester la requête ou OPTIONS OBS = MAX pour l’exécuter sur les
 données */

/*Critères de sélection des médicaments  à extraire du référentiel pharmacie (cf communiqué du 24 janvier 2012)*/
%LET codes = pha_cip_c13 in ( 3400933226558, 3400934744198, 3400926939939 ) ;
/* 		EXEMPLE 1 : Sélection par CIP 13
			pha_cip_c13 in (3400933226558,3400934744198,3400926939939)
		EXEMPLE 2 : Sélection par CIP 7 
			pha_prs_ide  in ( 3876282, 3517778, 3383272 ) 
		EXEMPLE 3 : Sélection par code ATC  
			pha_atc_c03 like 'A10' and not(pha_atc_c07='A10BX06') 
		EXEMPLE 4 : Sélection par principe actif
			pha_nom_pa like %METFORMINE%
		exemple 5 : Sélection par CIP 7 ou ATC 7*
			pha_prs_ide  in ( 3876282, 3517778, 3383272 ) or  pha_atc_C07  in ("J01AA02", "J01FA01" ,"J01MA01")	*/

/*liste des variables à conserver (Ne pas changer les préfixes p., pha.,  ir.)*/

/* variables de la table des prestations*/
%LET liste_var_prs =
		p.ben_nir_psa,	
		p.ben_rng_gem,
		p.exe_soi_dtd,		
		p.ben_res_dpt, 
		p.org_aff_ben,
		p.pse_spe_cod,
		p.psp_spe_cod
		/*et autres variables selon le besoin*/;

/* variables de la table d’actes affinée  = ici pharmacie ER_PHA_F*/
%LET liste_var_pha =
       	pha.pha_prs_c13 ,  
		pha.pha_prs_ide 
		/*et autres variables selon le besoin*/;

/* variables du référentiel  = ici référenciel pharmacie IR_PHA_R*/
%LET liste_var_ir =
		ir.pha_atc_c03,  
		ir.pha_atc_c07 
		/*et autres variables selon le besoin*/;

/*1ère année de la table ER_PRS_F*/
%LET an_DCIR = 2013 ; 

/*Facultatif: Pour extraire les remboursements de médicaments d'une population déjà constituée, et non de la population française entière
Attention: 1 ligne = 1 individu 
Variables obligatoires : BEN_NIR_PSA, BEN_RNG_GEM
Si vous n'avez pas de table de population au départ: %let table_pop=  , sinon =nom de la table*/
%LET table_pop = ;


/**********************************/
/********    MACRO OUTILS    ******/
/**********************************/

/* 9  cles techniques  */
%MACRO CLE_TEC ( p, d ) ;
&p..DCT_ORD_NUM = &d..DCT_ORD_NUM and
&p..FLX_DIS_DTD = &d..FLX_DIS_DTD and
&p..FLX_EMT_NUM = &d..FLX_EMT_NUM and
&p..FLX_EMT_ORD = &d..FLX_EMT_ORD and
&p..FLX_EMT_TYP = &d..FLX_EMT_TYP and
&p..FLX_TRT_DTD = &d..FLX_TRT_DTD and
&p..ORG_CLE_NUM = &d..ORG_CLE_NUM and
&p..prs_ORD_NUM = &d..prs_ORD_NUM and 
&p..REM_TYP_AFF = &d..REM_TYP_AFF
%MEND ;


/* Lorsque l’on a besoin de fusionner une table personnelle avec une table oracle, il faut créer une table personnelle dans ORAUSER, avant cela il faut s’assurer qu’une table avec le même nom n’existe pas déjà ;  le cas échéant il faut supprimer cette table existante */
%MACRO KILL_ORAUSER ( table ) ;
%IF %SYSFUNC( exist( ORAUSER.&table )) %THEN %DO ;
	PROC SQL;
	DROP TABLE ORAUSER.&table;
	QUIT; 
%END ;
%MEND ;





/***************************************************************/
/****  CHOIX DES MEDICAMENTS DANS LE REFERENTIEL PHARMACIE  ****/
/***************************************************************/

%KILL_ORAUSER( liste_medicaments ) ;

PROC SQL; 
	CREATE TABLE ORAUSER.liste_medicaments  AS
	SELECT	* /*ou variables à choisir*/
	FROM 	ORAVUE.IR_PHA_R
	WHERE 	&codes 
	;
QUIT ;




/***************************************************************/
/******      OPTION : CHOIX d'UNE TABLE de POPULATION     ******/
/***************************************************************/

%MACRO table_pop() ;
%IF &table_pop ne  %THEN %DO ;
	%PUT Table de population = &table_pop. ;
	%KILL_ORAUSER(pop) ;
	DATA orauser.pop ;
	SET &table_pop ;
	RUN ;
	%MACRO croisement_tab_pop() ;
		inner join orauser.pop pop
		on pop.ben_nir_psa=p.ben_nir_psa and pop.ben_rng_gem=p.ben_rng_gem
	%MEND ;
%END ;
%ELSE %DO ;
	%PUT no_table ;
	%MACRO croisement_tab_pop() ;
	%MEND ;
%END ;
%MEND ;
%table_pop ;





/***************************************************************/
/************        SELECTION DES PRESTATIONS       ***********/
/***************************************************************/ 
/*** Macro de création d'une table de prestations mensuelle ***/

%MACRO CREA_TABLE ( FLX_DIS_DTD ) ;
%PUT ### Mois de flux ( &FLX_DIS_DTD ) Début %SYSFUNC( datetime(),datetime. ) ;

/* Suppression de la table de prestations mensuelle si existante */
%IF %SYSFUNC ( exist( table_mois )) %THEN %DO ;
	PROC DELETE DATA = table_mois ;
	RUN ;
%END ;

PROC SQL; 
CREATE TABLE table_mois  as 
SELECT  	&liste_var_prs,  
		&liste_var_pha,
		&liste_var_ir,
		pha.pha_act_qsn
FROM    &table_prs as p	%croisement_tab_pop /*ne fait rien si pas de table de population à 
							croiser*/
		INNER JOIN &table_pha as pha
			ON %CLE_TEC (p,pha)
[bookmark: _GoBack]		INNER JOIN orauser.liste_medicaments as ir
			ON 	&CONDITION_JOINTURE_MEDICAMENTS
WHERE   p.flx_dis_dtd ="&FLX_DIS_DTD:0:0:0"dt
AND (p.exe_soi_dtd between "&DATE_EXE_DEB:0:0:0"dt and "&DATE_EXE_FIN:0:0:0"dt ) /* sélection de la période d’exécution du soin*/
AND (p.DPN_QLF <> 71 AND p.PRS_DPN_QLP <> 71) /*exclusion des soins externes
 transmis pour info*/
		AND (p.CPL_MAJ_TOP<2) ;                   /*exclusion des majorations*/

QUIT ;

%PUT ### CREA_TABLE ( &FLX_DIS_DTD ) Fin %SYSFUNC( datetime(),datetime. ) ;
%MEND CREA_TABLE ;

/***  Macro Compilation des tables mensuelles ****/

%MACRO COMPILATION() ;
%IF %SYSFUNC(exist( compil )) eq 0 %THEN %DO ;
proc datasets nolist memtype=data;
change table_mois=compil;
quit;
%END ;
%ELSE %DO;
proc append base=compil data=table_mois FORCE;run;
proc delete data=table_mois;run;
%END;
%MEND COMPILATION ;

/*** Macro de suppression des régularisations ***/

%MACRO REGULARISATIONS( table_sortie );

/*liste des variables (hors variables quantité) sans les suffixes*/
%let liste_var_regul=%qsysfunc(tranwrd(%quote(&liste_var_prs),p.,)) , %qsysfunc(tranwrd(%quote(&liste_var_pha),pha., )), %qsysfunc(tranwrd(%quote(&liste_var_ir),ir., ));
 
PROC SQL ;
CREATE TABLE &table_sortie as
SELECT	distinct	
			&liste_var_regul,
			sum(pha_act_qsn) as pha_qt
FROM	compil 
GROUP BY	&liste_var_regul
HAVING CALCULATED	pha_qt>0 ;
QUIT ;
%MEND REGULARISATIONS;

/*** Exécution des macros précédentes pour chaque mois de flux***/

%MACRO TAB_MEDICAMENTS( Date_DEB= ,Date_FIN= ,table_sortie= ) ;
 
/*DEFINITION DES PARAMETRES*/
%LET AN_DEB = %SUBSTR( &DATE_DEB,1,4 ) ;
%LET mois_deb = %SUBSTR( &DATE_DEB,5,2 ) ;

%LET AN_FIN = %SUBSTR( &DATE_FIN,1,4 ) ;
%LET mois_fin = %SUBSTR( &DATE_FIN,5,2 ) ;

/*dates de soin en format SAS*/
%LET EXE_DEB = %SYSFUNC( mdy( &mois_deb,1,&AN_DEB )) ;
%LET EXE_FIN0 = %SYSFUNC( mdy( &mois_fin,1,&an_fin )) ; 
%LET EXE_FIN = %EVAL( %SYSFUNC( intnx( month,&EXE_FIN0, 1 ))-1 ) ; /*pour aller jusqu'au dernier jour du mois de la date de fin donnée*/

/*dates de soin en format date9*/
%LET DATE_EXE_DEB = %SYSFUNC( putn( &EXE_DEB,date9. )) ;
%LET DATE_EXE_FIN = %SYSFUNC( putn( &EXE_FIN,date9. )) ;
 
 /*suppression de la table compilée des mois extraits si ce n'est pas la 1ère exécution*/
%IF %SYSFUNC(exist(compil)) %THEN %DO;
 	PROC DELETE DATA=compil;
	RUN ;
%END;

/*BOUCLE MENSUELLE DE SELECTION DES PRESTATIONS*/
%LET iterDeb = 1; 
%LET iterFin = %SYSFUNC( intck( month,&EXE_DEB,&EXE_FIN )) ;/*Nombre de mois entre le début et la fin des soins */

/*BOUCLE SUR LES MOIS*/
%DO mois=&iterDeb %TO &iterFin + 7 ;
	%LET FLX_DIS_DTD = %SYSFUNC( intnx( MONTH,&EXE_DEB,&mois ), date9. ) ;
	%LET an = %SYSFUNC( year( %SYSFUNC( intnx( MONTH,&EXE_DEB,&mois-1 )))) ;

/*jointure par code CIP13 ou code CIP7 selon l'année*/
	%IF  &an<2014 %THEN %LET CONDITION_JOINTURE_MEDICAMENTS=(pha.pha_prs_ide=ir.pha_prs_ide);
	%ELSE %LET CONDITION_JOINTURE_MEDICAMENTS=(pha.pha_prs_c13=ir.pha_cip_c13);
		
				 
	/*DCIR archivé*/
	%IF &an<&an_DCIR %THEN %DO;
		%LET table_prs = oravue.er_prs_f_&an ;
		%LET table_pha = oravue.er_pha_f_&an ;
	%END;
	/*DCIR courant*/
	%IF  &AN >= &an_DCIR %THEN %DO;
		%LET table_prs = oravue.er_prs_f  ;
		%LET table_pha = oravue.er_pha_f  ;
	%END;

	/*affichage dans le journal*/
	%PUT "Valeurs des paramètres" ;
	%PUT DATE_EXE_DEB = &DATE_EXE_DEB ;
	%PUT DATE_EXE_FIN = &DATE_EXE_FIN ;
	%PUT an = &an ;
	%PUT &table_prs ;
	%PUT FLX_DIS_DTD = &FLX_DIS_DTD ;
%PUT CONDITION_JOINTURE_MEDICAMENTS=&CONDITION_JOINTURE_MEDICAMENTS;
	 %CREA_TABLE(&FLX_DIS_DTD) ;
	 %COMPILATION() ;			
%END ;
     %REGULARISATIONS( &table_sortie ) ;
%MEND TAB_MEDICAMENTS;

/****************************************************************************/
/*lancement de la macro de sélection des prestations                                                                       */
/****************************************************************************/
OPTIONS MPRINT ;
 %TAB_MEDICAMENTS(	 Date_DEB = 201401,Date_FIN = 201404,table_sortie = table_medicaments ) ;

/*********************************************************************************/
/************                   SELECTION DE DONNEES SOCIODEMOGRAPHIQUES                         *******/
/************                    DANS LES REFERENTIELS DES BENEFICIAIRES                                     *******/
/*********************************************************************************/
/* DONNEES en entrée 
				&table_entree à définir
				ORAVUE.IR_BEN_R
				ORAVUE.IR_BEN_R_ARC
/* DONNEES en sortie
				&table_sortie à définir
/* PLAN du PROGRAMME ;
	  			- Définition par l'utilisateur des paramètres suivants:
					* table_entree= table de population dont on veut connaître
 les caractéristiques
					* table_sortie= table après croisement avec les référentiels
		
	  			- Liste des variables d'intérêt dans les référentiels
				- Copie de la table entrée dans ORAUSER
	  			- Croisement des données avec IR_BEN_R
				- Croisement des données avec IR_BEN_R_ARC (référentiel archivé) 
	  			- Compilation des deux tables
				- Conservation de la dernière information mise à jour pour chaque
 BEN_NIR_PSA/BEN_RNG_GEM
	 
/*********************************************************************************/
/*********************************************************************/
/*****     CROISEMENT AVEC LE REFERENTIEL DES BENEFICIAIRES                   ********/
/*********************************************************************/ 

%let table_entree=table_medicaments;
%let table_sortie=table_IR;

/*copie de la table dans ORAUSER*/
%KILL_ORAUSER( &table_entree) ;

PROC SQL; 
	CREATE TABLE ORAUSER.&table_entree AS
	SELECT	* /*ou variables à choisir, au minimum BEN_NIR_PSA, BEN_RNG_GEM*/
	FROM 	&table_entree
	;
QUIT ;


/*Croisement avec les référentiels
- IR_BEN_R pour les individus ayant consommé au moins 1 fois depuis le 1er janvier 2013
- IR_BEN_R_ARC pour les individus dont les derniers remboursements sont antérieurs à 2013*/

%let liste_var_IR_BEN =
	ben_nai_moi,
	ben_sex_cod,
	ben_dcd_dte, /*sur profil 108*/
	ben_nir_ano,
	ben_idt_ano,
	ben_dte_maj,
	max_trt_dtd
	;

proc sql ;

/*Croisement avec IR_BEN*/
create table tab_IR_BEN (compress=yes) as 
select a.*,&liste_var_IR_BEN
from   ORAUSER.&table_entree a , ORAVUE.IR_BEN_R  b 
where a.ben_nir_psa = b.ben_nir_psa and
	 a.ben_rng_gem = b.ben_rng_gem
;

/*Croisement avec IR_BEN_R_ARC*/
create table tab_IR_BEN_ARC (compress=yes) as 
select  a.*, &liste_var_IR_BEN
from   ORAUSER.&table_entree a  , ORAVUE.IR_BEN_R_ARC  b 

where a.ben_nir_psa = b.ben_nir_psa and
	  a.ben_rng_gem = b.ben_rng_gem;
quit;


/*Compilation des 2 tables*/
data tab_IR;set tab_IR_BEN (in=a) tab_IR_BEN_ARC(in=b);
if b then source='IR_BEN_ARC';if a then source='IR_BEN';run;


/*Conservation de l'information la plus récente par ben_nir_psa/ben_rng_gem*/
proc sort data=tab_IR;by ben_nir_psa ben_rng_gem descending ben_dte_maj descending max_trt_dtd;run;

proc sort data=tab_IR nodupkey out=&table_sortie; by ben_nir_psa ben_rng_gem;run;

